NAG Library Routine Document

E01BAF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

E01BAF determines a cubic spline interpolant to a given set of data.

2 Specification

```
SUBROUTINE E01BAF(M, X, Y, LAMDA, C, LCK, WRK, LWRK, IFAIL)
INTEGER           M, LCK, LWRK, IFAIL
double precision  X(M), Y(M), LAMDA(LCK), C(LCK), WRK(LWRK)
```

3 Description

E01BAF determines a cubic spline $s(x)$, defined in the range $x_1 \leq x \leq x_m$, which interpolates (passes exactly through) the set of data points (x_i, y_i), for $i = 1, 2, \ldots, m$, where $m \geq 4$ and $x_1 < x_2 < \cdots < x_m$. Unlike some other spline interpolation algorithms, derivative end conditions are not imposed. The spline interpolant chosen has $m - 4$ interior knots $\lambda_5, \lambda_6, \ldots, \lambda_m$, which are set to the values of $x_3, x_4, \ldots, x_{m-2}$ respectively. This spline is represented in its B-spline form (see Cox (1975)):

$$s(x) = \sum_{i=1}^{m} c_i N_i(x),$$

where $N_i(x)$ denotes the normalized B-spline of degree 3, defined upon the knots $\lambda_i, \lambda_{i+1}, \ldots, \lambda_{i+4}$, and c_i denotes its coefficient, whose value is to be determined by the routine.

The use of B-splines requires eight additional knots λ_1, λ_2, λ_3, λ_4, λ_{m+1}, λ_{m+2}, λ_{m+3} and λ_{m+4} to be specified; E01BAF sets the first four of these to x_1 and the last four to x_m.

The algorithm for determining the coefficients is as described in Cox (1975) except that QR factorization is used instead of LU decomposition. The implementation of the algorithm involves setting up appropriate information for the related routine E02BAF followed by a call of that routine. (See E02BAF for further details.)

Values of the spline interpolant, or of its derivatives or definite integral, can subsequently be computed as detailed in Section 8.

4 References

Cox M G (1975) An algorithm for spline interpolation *J. Inst. Math. Appl.* **15** 95–108

Cox M G (1977) A survey of numerical methods for data and function approximation *The State of the Art in Numerical Analysis* (ed D A H Jacobs) 627–668 Academic Press

5 Parameters

1: M – INTEGER *Input*

On entry: m, the number of data points.

Constraint: M ≥ 4.

2: X(M) – ***double precision*** array *Input*

On entry: X(i) must be set to x_i, the ith data value of the independent variable x, for $i = 1, 2, \ldots, m$.

Constraint: X(i) < X($i + 1$), for $i = 1, 2, \ldots, \mathrm{M} - 1$.

3: Y(M) – ***double precision*** array *Input*

On entry: Y(i) must be set to y_i, the ith data value of the dependent variable y, for $i = 1, 2, \ldots, m$.

4: LAMDA(LCK) – ***double precision*** array *Output*

On exit: the value of λ_i, the ith knot, for $i = 1, 2, \ldots, m + 4$.

5: C(LCK) – ***double precision*** array *Output*

On exit: the coefficient c_i of the B-spline $N_i(x)$, for $i = 1, 2, \ldots, m$. The remaining elements of the array are not used.

6: LCK – INTEGER *Input*

On entry: the dimension of the arrays LAMDA and C as declared in the (sub)program from which E01BAF is called.

Constraint: LCK $\geq$ M + 4.

7: WRK(LWRK) – ***double precision*** array *Workspace*

8: LWRK – INTEGER *Input*

On entry: the dimension of the array WRK as declared in the (sub)program from which E01BAF is called.

Constraint: LWRK $\geq 6 \times$ M + 16.

9: IFAIL – INTEGER *Input/Output*

On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.

On exit: IFAIL = 0 unless the routine detects an error (see Section 6).

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this parameter, the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

6 Error Indicators and Warnings

If on entry IFAIL = 0 or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL = 1

On entry, M < 4,
or LCK < M + 4,
or LWRK < 6 × M + 16.

IFAIL = 2

The X-values fail to satisfy the condition

X(1) < X(2) < X(3) < ⋯ < X(M).

7 Accuracy

The rounding errors incurred are such that the computed spline is an exact interpolant for a slightly perturbed set of ordinates $y_i + \delta y_i$. The ratio of the root-mean-square value of the δy_i to that of the y_i is no greater than a small multiple of the relative ***machine precision***.

8 Further Comments

The time taken by E01BAF is approximately proportional to m.

All the x_i are used as knot positions except x_2 and x_{m-1}. This choice of knots (see Cox (1977)) means that $s(x)$ is composed of $m-3$ cubic arcs as follows. If $m = 4$, there is just a single arc space spanning the whole interval x_1 to x_4. If $m \geq 5$, the first and last arcs span the intervals x_1 to x_3 and x_{m-2} to x_m respectively. Additionally if $m \geq 6$, the ith arc, for $i = 2, 3, \ldots, m-4$, spans the interval x_{i+1} to x_{i+2}.

After the call

```
CALL E01BAF (M, X, Y, LAMDA, C, LCK, WRK, LWRK, IFAIL)
```

the following operations may be carried out on the interpolant $s(x)$.

The value of $s(x)$ at $x = \mathrm{X}$ can be provided in the ***double precision*** variable S by the call

```
CALL E02BBF (M+4, LAMDA, C, X, S, IFAIL)
```

(see E02BBF).

The values of $s(x)$ and its first three derivatives at $x = \mathrm{X}$ can be provided in the ***double precision*** array S of dimension 4, by the call

```
CALL E02BCF (M+4, LAMDA, C, X, LEFT, S, IFAIL)
```

(see E02BCF).

Here LEFT must specify whether the left- or right-hand value of the third derivative is required (see E02BCF for details).

The value of the integral of $s(x)$ over the range x_1 to x_m can be provided in the ***double precision*** variable DINT by

```
CALL E02BDF (M+4, LAMDA, C, DINT, IFAIL)
```

(see E02BDF).

9 Example

This example sets up data from 7 values of the exponential function in the interval 0 to 1. E01BAF is then called to compute a spline interpolant to these data.

The spline is evaluated by E02BBF, at the data points and at points halfway between each adjacent pair of data points, and the spline values and the values of e^x are printed out.

9.1 Program Text

```
*     E01BAF Example Program Text
*     Mark 14 Revised. NAG Copyright 1989.
*     .. Parameters ..
      INTEGER          M, LCK, LWRK
      PARAMETER        (M=7,LCK=M+4,LWRK=6*M+16)
      INTEGER          NOUT
      PARAMETER        (NOUT=6)
*     .. Local Scalars ..
      DOUBLE PRECISION FIT, XARG
      INTEGER          I, IFAIL, J, R
*     .. Local Arrays ..
      DOUBLE PRECISION C(LCK), LAMDA(LCK), WRK(LWRK), X(M), Y(M)
*     .. External Subroutines ..
      EXTERNAL         E01BAF, E02BBF
*     .. Intrinsic Functions ..
```

```
      INTRINSIC        EXP
*     .. Data statements ..
      DATA             (X(I),I=1,M)/0.0D0, 0.2D0, 0.4D0, 0.6D0, 0.75D0,
     +                 0.9D0, 1.0D0/
*     .. Executable Statements ..
      WRITE (NOUT,*) 'E01BAF Example Program Results'
      DO 20 I = 1, M
         Y(I) = EXP(X(I))
   20 CONTINUE
      IFAIL = 1
*
      CALL E01BAF(M,X,Y,LAMDA,C,LCK,WRK,LWRK,IFAIL)
*
      IF (IFAIL.EQ.0) THEN
         WRITE (NOUT,*)
         WRITE (NOUT,*) '   J    Knot LAMDA(J+2)   B-spline coeff C(J)'
         WRITE (NOUT,*)
         J = 1
         WRITE (NOUT,99998) J, C(1)
         DO 40 J = 2, M - 1
            WRITE (NOUT,99999) J, LAMDA(J+2), C(J)
   40    CONTINUE
         WRITE (NOUT,99998) M, C(M)
         WRITE (NOUT,*)
         WRITE (NOUT,*)
     +      '   R        Abscissa            Ordinate             Spline'
         WRITE (NOUT,*)
         DO 60 R = 1, M
            IFAIL = 1
*
            CALL E02BBF(M+4,LAMDA,C,X(R),FIT,IFAIL)
*
            WRITE (NOUT,99999) R, X(R), Y(R), FIT
            IF (R.LT.M) THEN
               XARG = 0.5D0*(X(R)+X(R+1))
*
               CALL E02BBF(M+4,LAMDA,C,XARG,FIT,IFAIL)
*
               WRITE (NOUT,99997) XARG, FIT
            END IF
   60    CONTINUE
      ELSE
         WRITE (NOUT,99996) IFAIL
      END IF
*
99999 FORMAT (1X,I4,F15.4,2F20.4)
99998 FORMAT (1X,I4,F35.4)
99997 FORMAT (1X,F19.4,F40.4)
99996 FORMAT (1X,' ** E01BAF returned with IFAIL = ',I5)
      END
```

9.2 Program Data

None.

9.3 Program Results

```
 E01BAF Example Program Results

    J    Knot LAMDA(J+2)   B-spline coeff C(J)

    1                           1.0000
    2         0.0000            1.1336
    3         0.4000            1.3726
    4         0.6000            1.7827
    5         0.7500            2.1744
    6         1.0000            2.4918
    7                           2.7183

    R        Abscissa           Ordinate             Spline
```

```
 1        0.0000          1.0000          1.0000
          0.1000                          1.1052
 2        0.2000          1.2214          1.2214
          0.3000                          1.3498
 3        0.4000          1.4918          1.4918
          0.5000                          1.6487
 4        0.6000          1.8221          1.8221
          0.6750                          1.9640
 5        0.7500          2.1170          2.1170
          0.8250                          2.2819
 6        0.9000          2.4596          2.4596
          0.9500                          2.5857
 7        1.0000          2.7183          2.7183
```